




Apple /// Computer Information



DOCUMENT NAME	#
<i>BUSINESS BASIC PROGRAM TO HELP CLEAN DISK DRIVES</i>	<i>93</i>

Ex Libris David T. Craig

8/29/89 2:50 PM

Disk # 3:apple.DISK CLEANER.TEXT

Page 1

Title: A program to help clean disk drives -
 Created by: ED GOODING on: 10/29/1988 21:21:29

BUSINESS BASIC

```

1  REM      **** A DISK DRIVE CLEANER PROGRAM ****
10  ON ERR GOTO 500
11  VPOS=13:HPOS=6:
20  TEXT:HOME:VPOS=1:HPOS=30:PRINT"DISK CLEANING PROGRAM"
30  VPOS=6:HPOS=6:PRINT CHR$(2);
40  PRINT"This program allows you to use one of the commercially available"
50  PRINT"head cleaner disks with your Apple ///. Input how many drives you"
60  PRINT"you will be cleaning (up to four) and insert the cleaner disk in"
70  PRINT"the appropriate drive at the prompt.":TEXT
105  ON ERR IF ERR=254 OR ERR=5 THEN PRINT CHR$(7);:GOTO 110:ELSE GOTO 500
110  VPOS=15:HPOS=6:PRINT CHR$(31);
111  VPOS=15:HPOS=6:INPUT"Total number of drives to be cleaned: ";total%
112  IF total%>4 OR total%<1 THEN PRINT CHR$(7);:GOTO 110
113  VPOS=17:HPOS=6:PRINT CHR$(31);
114  VPOS=17:HPOS=6:INPUT"What drive do you want to start with?: ";start%
115  IF start%>4 OR start%<1 THEN PRINT CHR$(7);:GOTO 113
116  whoa%=start%+total%-1
118  ON ERR GOTO 500
119  WINDOW 6,22 TO 80,24
120  FOR count1%=start% TO whoa%
130  HOME:OFF KBD
140  PRINT"Insert the cleaning disk in drive .D";count1%:PRINT
150  PRINT"Press 'RETURN' to begin, 'ESCAPE' to end.";
160  GET g$:IF g$=CHR$(27) THEN 300
170  IF g$<>CHR$(13) THEN PRINT CHR$(7);:GOTO 160
180  HOME:PRINT
185  ON KBD GOTO 600
190  PRINT"Cleaning drive .D";count1%;", 'ESCAPE' to abort. Please wait";
200  FOR count2%=1 TO 10
210  ON ERR IF ERR=32 GOTO 230:ELSE GOTO 500
220  CATALOG(".D"+CONV$(count1%))
225  HOME:PRINT:INVERSE:PRINT"NOT A CLEANING DISK. PRESS 'RETURN'."
226  NORMAL:GET g$:GOTO 130
230  PRINT".";
235  ON ERR GOTO 500
240  NEXT
250  NEXT
300  HOME:TEXT:OFF KBD
310  VPOS=18:HPOS=6:PRINT CHR$(2);
320  INVERSE:PRINT"'RETURN' to repeat, 'ESCAPE' to exit";:NORMAL
330  GET g$:IF g$=CHR$(13) THEN 10
340  IF g$<>CHR$(27) THEN PRINT CHR$(7);:GOTO 330
350  TEXT:HOME:END
500  TEXT:VPOS=23:HPOS=1:INVERSE:PRINT"Error #"; ERR;" in Line #"; ERRLIN;
510  NORMAL:END
600  IF KBD=27 THEN POP:GOTO 300
610  ON KBD GOTO 600
620  RETURN

```